# Project     : QKeyboardStatistic
# Author      : Durachenko Aleksey V. <durachenko.aleksey@gmail.com>, 2011
# Licence     : GNU GENERAL PUBLIC LICENSE Version 2

DESTDIR         = bin
OBJECTS_DIR     = build_shared/obj
MOC_DIR         = build_shared/moc
UI_DIR          = build_shared/ui
RCC_DIR         = build_shared/rcc

TARGET          = QKeyboardStatistic
VERSION         = 0.1

TEMPLATE        = app
CONFIG         += debug_and_release
QT             += core gui sql

DEFINES        +=

SOURCES        +=           \
    src/main.cpp            \
    src/cmainwindow.cpp     \
    src/chookthread.cpp     \
    src/cstatistic.cpp      \
    src/cdatabase.cpp       \
    src/csettingsdialog.cpp \
    src/caboutdialog.cpp

HEADERS        +=           \
    src/version.h           \
    src/global.h            \
    src/cmainwindow.h       \
    src/chookthread.h       \
    src/cstatistic.h        \
    src/cdatabase.h         \
    src/csettingsdialog.h   \
    src/caboutdialog.h


FORMS          +=           \
    ui/mainwindow.ui        \
    ui/settingsdialog.ui    \
    ui/aboutdialog.ui

RESOURCES      +=           \
    rc/rc.qrc

TRANSLATIONS    =

INCLUDEPATH    +=           \
    src

build_pass:CONFIG(debug, debug|release) {
    TARGET      = $$join(TARGET,,,d)
    OBJECTS_DIR = $$join(OBJECTS_DIR,,,d)
    MOC_DIR     = $$join(MOC_DIR,,,d)
    UI_DIR      = $$join(UI_DIR,,,d)
    RCC_DIR     = $$join(RCC_DIR,,,d)

    win32 {
        RC_FILE += rc/rc.rc
        LIBS    +=
    }

    unix {
        LIBS    +=
    }
}

build_pass:CONFIG(release, debug|release) {
    win32 {
        RC_FILE += rc/rc.rc
        LIBS    +=
    }

    unix {
        LIBS    +=
    }
}
